
PROIECTAREA SISTEMELOR DIGITALE 
Masterat ICCP - an 1 

 
LABORATOR 3 

 
INSTRUCłIUNI SECVENłIALE ÎN LIMBAJUL VHDL 

Partea 1 
 
 

O descriere în limbajul VHDL are două domenii: un domeniu secvenŃial şi un 
domeniu concurent. Domeniul secvenŃial este reprezentat de un proces sau subprogram care 
conŃine instrucŃiuni secvenŃiale. Aceste instrucŃiuni sunt executate în ordinea în care apar în 
cadrul procesului sau subprogramului, ca şi în cazul limbajelor de programare. Domeniul 
concurent este reprezentat de o arhitectură, care conŃine procese, apeluri concurente de 
proceduri, asignări concurente ale semnalelor şi instanŃieri de componente. Toate activităŃile 
descrise de acestea au loc simultan. 

În această lucrare de laborator se prezintă formatul şi modul de utilizare al unor 
instrucŃiuni secvenŃiale. Pentru exemplificare sunt descrise unele componente combinaŃionale 
şi secvenŃiale de bază, ca de exemplu: multiplexoare, codificatoare, decodificatoare, 
numaratoare. 

 
3.1. InstrucŃiuni secvenŃiale 

 
În această secŃiune se descriu mai întâi unele aspecte legate de procese, cum sunt 

modul de specificare a unui proces, execuŃia unui proces, instrucŃiunea wait şi deosebirea 
dintre procesele combinaŃionale şi cele secvenŃiale. Sunt prezentate apoi instrucŃiunile 
secvenŃiale care pot apare într-un proces sau subprogram: asignarea secvenŃială a semnalelor, 
asignarea variabilelor, instrucŃiunea if.  

 
3.1.1. Procese 
Un proces este o secvenŃă de instrucŃiuni care sunt executate în ordinea specificată. 

DeclaraŃia unui proces delimitează un domeniu secvenŃial al arhitecturii în care apare 
declaraŃia. Procesele se utilizează pentru descrieri funcŃionale. 
 

3.1.1.1. Structura şi execuŃia unui proces 
 

Un proces poate apare oriunde în corpul unei arhitecturi (partea care începe după 
cuvântul cheie begin). Structura de bază a declaraŃiei unui proces este următoarea: 

[nume:] process [(listă_de_sensibilitate)] 
[declaraŃii_de_tipuri] 
[declaraŃii_de_constante] 
[declaraŃii_de_variabile] 
[declaraŃii_de_subprograme] 

begin 
instrucŃiuni_secvenŃiale 

end process [nume]; 

DeclaraŃia unui proces este cuprinsă între cuvintele cheie process şi end process. 
Unui proces i se poate asigna în mod opŃional un nume pentru identificarea mai uşoară a 
procesului în textul sursă. Numele este un identificator şi trebuie urmat de caracterul ':' (cu rol 
de etichetă). Acest nume este util şi pentru simulare, de exemplu, pentru setarea unui punct de 
întrerupere a execuŃiei simulării. Numele poate fi repetat la sfârşitul declaraŃiei, după 
cuvintele cheie end process. 

Lista de sensibilitate (opŃională) este lista semnalelor la a căror modificare procesul 
este “sensibil”. Producerea unui eveniment asupra unuia din semnalele specificate în lista de 
sensibilitate va determina execuŃia instrucŃiunilor secvenŃiale din cadrul procesului, similar cu 
instrucŃiunile dintr-un program obişnuit. Spre deosebire de un limbaj de programare, în 
limbajul VHDL clauza end process nu specifică terminarea execuŃiei procesului. 

Procesul va fi executat într-un ciclu infinit, iar în cazul în care se specifică o listă de 
sensibilitate, procesul va fi doar suspendat după execuŃia ultimei instrucŃiuni, până la 
producerea unui nou eveniment asupra semnalelor din listă. Se menŃionează că un eveniment 
are loc numai la modificarea valorii unui semnal. Astfel, asignarea aceleiaşi valori la un 
semnal nu reprezintă un eveniment. 

În cazul în care lista de sensibilitate lipseşte, procesul va fi executat în mod continuu. 
În acest caz, procesul trebuie să conŃină o instrucŃiune wait pentru a determina suspendarea 
procesului şi activarea acestuia la apariŃia unui eveniment sau îndeplinirea unei condiŃii. În 
cazul în care lista de sensibilitate este prezentă, procesul nu poate conŃine instrucŃiuni wait.  

Partea declarativă a procesului este cuprinsă între cuvintele cheie process şi begin. În 
această parte se pot declara tipuri, constante, variabile şi subprograme (proceduri şi funcŃii) 
care sunt locale procesului. Elementele declarate se pot utiliza deci numai în interiorul 
procesului. 

ObservaŃie 
• În cadrul unui proces nu pot fi declarate semnale, ci numai constante şi variabile. 

 
Partea de instrucŃiuni a procesului începe după cuvântul cheie begin. Această parte 

conŃine instrucŃiunile care vor fi executate la fiecare activare a procesului. Nu este permisă 
utilizarea instrucŃiunilor concurente în interiorul unui proces. 

În Exemplul 1 se prezintă declaraŃia unui proces simplu format dintr-o singură 
instrucŃiune de asignare secvenŃială. 

Exemplul 1: 
proc1: process (a, b, c) 
begin 

x <= a and b and c; 
end process proc1; 

 
3.1.1.2. Procese cu liste de sensibilitate incomplete 

 
Unele sisteme de sinteză nu verifică listele de sensibilitate ale proceselor. Aceste 

sisteme presupun că toate semnalele din partea dreaptă a asignărilor secvenŃiale la semnale se 
află în lista de sensibilitate. Astfel, aceste sisteme vor interpreta cele două procese din 
Exemplul 2 ca fiind identice. 

Exemplul 2: 
proc1: process (a, b, c) 
begin 

x <= a and b and c; 
end process proc1; 

proc2: process (a, b) 
begin 

x <= a and b and c; 
end process proc2; 

 
Toate sistemele de sinteză vor interpreta procesul proc1 ca o poartă ŞI cu 3 intrări. 

Unele sisteme de sinteză vor interpreta procesul proc2 de asemenea ca o poartă ŞI cu 3 intrări, 
chiar dacă la simularea acestui cod procesul nu se va comporta ca atare. La simulare, o 
modificare a valorii semnalului a sau b va determina execuŃia procesului, iar valoarea funcŃiei 
ŞI logic între semnalele a, b şi c se va asigna semnalului x. Totuşi, dacă se modifică valoarea 
semnalului c, procesul nu este executat, iar semnalul x nu este actualizat. 

Deoarece nu este clar modul în care un sistem de sinteză ar trebui să genereze un 
circuit pentru care o tranziŃie a semnalului c nu determină o modificare a semnalului x, dar o 
modificare a semnalelor a sau b determină actualizarea semnalului x cu valoarea funcŃiei ŞI 
logic între semnalele a, b şi c, există următoarele alternative pentru sistemele de sinteză: 

• Interpretarea procesului proc2 în mod identic cu procesul proc1 (cu o listă de 
sensibilitate care cuprinde toate semnalele din partea dreaptă a instrucŃiunilor de 
asignare a semnalelor din cadrul procesului); 

• Indicarea unei erori la compilare, specificând faptul că nu se poate realiza sinteza 
procesului fără o listă de sensibilitate completă. 

 
A doua variantă este preferabilă, deoarece proiectantul va trebui să modifice codul 

sursă astfel încât funcŃionarea circuitului generat să fie aceeaşi cu rezultatul simulării 
funcŃionale a codului sursă. 

Deşi din punct de vedere sintactic procesele pot fi declarate fără o listă de sensibilitate 
şi fără o instrucŃiune wait, asemenea procese nu vor fi suspendate niciodată. De aceea, dacă se 
va simula un asemenea proces, timpul de simulare nu va avansa, deoarece faza de iniŃializare, 
în care toate procesele sunt executate până când acestea sunt suspendate, nu se va termina 
niciodată. 
 

3.1.1.3. InstrucŃiunea wait 
 
În locul unei liste de sensibilitate, un proces poate conŃine o instrucŃiune wait. 

Utilizarea unei instrucŃiuni wait are două scopuri: 
• Suspendarea execuŃiei unui proces; 
• Specificarea condiŃiei care va determina activarea procesului suspendat. 

 
La întâlnirea unei instrucŃiuni wait, procesul în care apare această instrucŃiune este 

suspendat. Atunci când se îndeplineşte condiŃia specificată în cadrul instrucŃiunii wait, 
procesul este activat şi se execută instrucŃiunile acestuia până când se întâlneşte din nou 
instrucŃiunea wait. Limbajul VHDL permite ca un proces să conŃină mai multe instrucŃiuni 
wait. Atunci când se utilizează pentru modelarea logicii combinaŃionale în vederea sintezei, 
un proces poate conŃine însă o singură instrucŃiune wait. 

Dacă un proces conŃine o instrucŃiune wait, nu poate conŃine o listă de sensibilitate. 
Procesul din Exemplul 3, care conŃine o instrucŃiune wait explicită, este echivalent cu 
procesul din Exemplul 1, care conŃine o listă de sensibilitate. Ambele procese se vor executa 
atunci când apare o modificare a valorii semnalelor a, b sau c. 

Exemplul 3: 
proc3: process 
begin 

x <= a and b and c; 
wait on a, b, c; 
end process proc3; 

 
Există trei forme ale instrucŃiunii wait: 

wait on listă_de_sensibilitate; 
wait until expresie_condiŃională; 
wait for expresie_de_timp; 

 
InstrucŃiunea wait on a fost ilustrată în exemplele precedente. InstrucŃiunea wait until 

suspendă un proces până când condiŃia specificată devine adevărată datorită modificării unuia 
din semnalele care apar în expresia condiŃională. Se menŃionează că dacă nici un semnal din 
această expresie nu se modifică, procesul nu va fi activat, chiar dacă expresia condiŃională 
este adevărată. Exemplele următoare prezintă mai multe forme ale instrucŃiunii wait until: 

wait until semnal = valoare; 
wait until semnal’event and semnal = valoare; 
wait until not semnal’stable and semnal = valoare; 

unde semnal este numele unui semnal, iar valoare este valoarea care se testează. Dacă 
semnalul este de tip bit, atunci pentru valoarea '1' se aşteaptă frontul crescător al semnalului, 
iar pentru '0' frontul descrescător. 

InstrucŃiunea wait until se poate utiliza pentru implementarea unei funcŃionări 
sincrone. În mod obişnuit, semnalul testat este un semnal de ceas. De exemplu, aşteptarea 
frontului crescător al unui semnal de ceas se poate exprima în următoarele moduri: 

wait until clk = '1'; 
wait until clk'event and clk = '1'; 
wait until not clk'stable and clk = '1'; 

 
Pentru descrierile destinate sintezei, instrucŃiunea wait until trebuie să fie prima din 

cadrul procesului. Din această cauză, logica sincronă descrisă cu o instrucŃiune wait until nu 
poate fi resetată în mod asincron. 

InstrucŃiunea wait for permite suspendarea execuŃiei unui proces pentru un timp 
specificat, de exemplu: 

wait for 10 ns; 
 
Se pot combina mai multe condiŃii ale instrucŃiunii wait într-o condiŃie mixtă. În 

Exemplul 4, procesul proc4 va fi activat la modificarea valorii unuia din semnalele a sau b, 
dar numai atunci când valoarea semnalului clk este '1'. 

Exemplul 4: 
proc4: process 
begin 

wait on a, b until clk = '1'; 
... 
end process proc4; 

 
De obicei, instrucŃiunea wait apare fie la începutul unui proces, fie la sfârşitul 

acestuia. În Exemplul 3, instrucŃiunea wait apare la sfârşitul procesului. S-a menŃionat că 

această formă a procesului este echivalentă cu forma care conŃine o listă de sensibilitate. 
Această echivalenŃă se datorează modului în care se execută simularea unui model. În faza de 
iniŃializare a simulării, se execută toate procesele modelului. Dacă procesul conŃine o listă de 
sensibilitate, toate instrucŃiunile procesului vor fi executate o singură dată. În forma 
procesului care conŃine o instrucŃiune wait şi aceasta apare la sfârşit, în faza de iniŃializare se 
execută o singură dată toate instrucŃiunile până la instrucŃiunea wait, ca şi în cazul formei care 
conŃine o listă de sensibilitate. 

În cazul în care instrucŃiunea wait apare la începutul procesului, simularea se va 
executa în mod diferit, deoarece în faza de iniŃializare procesul va fi suspendat fără a se 
executa nici o instrucŃiune a acestuia. Deci, un proces cu instrucŃiunea wait plasată la început 
nu este echivalent cu un proces care conŃine o listă de sensibilitate. 

Deoarece faza de iniŃializare a simulării nu are echivalent hardware, nu vor exista 
diferenŃe la sinteza celor două procese cu instrucŃiunea wait plasată la sfârşit, respectiv la 
început. 

Totuşi, din cauza diferenŃei între simulare şi sinteză în ceea ce priveşte faza de 
iniŃializare a simulării, pot exista diferenŃe între funcŃionarea modelului la simulare şi 
funcŃionarea circuitului rezultat prin sinteză. Este posibil ca modelul care funcŃionează corect 
la simulare să fie sintetizat într-un circuit a cărui funcŃionare este eronată. 
 

3.1.1.4. Procese combinaŃionale şi procese secvenŃiale 
 
Atât procesele combinaŃionale, cât şi procesele secvenŃiale sunt interpretate în acelaşi 

fel, singura deosebire dintre acestea fiind că la procesele secvenŃiale semnalele de ieşire se 
înscriu în bistabile. Un process combinaŃional simplu este prezentat în Exemplul 5. 

Exemplul 5: 
proc5: process 
begin 

wait on a, b; 
z <= a and b; 

end process proc5; 
 

Acest proces va fi implementat prin sinteză ca o simplă poartă ŞI cu două intrări. 
Pentru ca un proces să modeleze un circuit combinaŃional, acesta trebuie să conŃină în lista de 
sensibilitate toate semnalele care sunt intrări ale procesului. Cu alte cuvinte, procesul trebuie 
reevaluat de fiecare dată când una din intrările circuitului pe care îl modelează se modifică. În 
acest fel se modelează în mod corect un circuit combinaŃional. 

Dacă un proces nu este sensibil la toate intrările sale şi nu este un proces secvenŃial, 
atunci nu poate fi sintetizat, deoarece nu există un echivalent hardware al unui asemenea 
proces. Nu toate sistemele de sinteză impun o asemenea regulă, astfel încât trebuie acordată 
atenŃie la scrierea proceselor combinaŃionale pentru a nu introduce erori. Asemenea erori vor 
avea ca efect diferenŃe subtile între modelul simulat şi circuitul obŃinut prin sinteză, deoarece 
un proces non-combinaŃional este interpretat de sistemul de sinteză ca un circuit 
combinaŃional. 

Dacă un proces conŃine o instrucŃiune wait sau o instrucŃiune if semnal'event, 
procesul va fi interpretat ca un proces secvenŃial. Astfel, procesul din Exemplul 6 va fi 
interpretat ca un proces secvenŃial. 

Exemplul 6: 
proc6: process 
begin 

wait until clk = '1'; 
z <= a and b; 

end process proc6; 
 

3.1.2. InstrucŃiunea secvenŃială de asignare a semnalelor 
 
Semnalele reprezintă interfaŃa dintre domeniul concurent al unui model în limbajul 

VHDL şi domeniul secvenŃial din cadrul unui proces. Un model VHDL este format dintr-un 
număr de procese care comunică între ele prin intermediul semnalelor. La simulare, întreaga 
ierarhie din cadrul modelului este eliminată, rămânând numai procesele şi semnalele. 
Simulatorul execută în mod alternativ actualizarea valorii unor semnale şi rularea proceselor 
activate de modificarea valorii semnalelor aflate în listele lor de sensibilitate. 
 

3.1.2.1. ExecuŃia instrucŃiunii secvenŃiale de asignare 
 

Asignarea valorilor la semnale se poate realiza printr-o instrucŃiune secvenŃială sau o 
instrucŃiune concurentă. InstrucŃiunea secvenŃială poate apare doar în interiorul unui proces, 
iar cea concurentă poate apare doar în exteriorul proceselor.  

InstrucŃiunea secvenŃială de asignare are o singură formă, cea simplă, care este o 
asignare necondiŃionată. InstrucŃiunea concurentă are, pe lângă forma sa simplă, încă două 
forme: asignarea condiŃională şi asignarea selectivă.  

InstrucŃiunea secvenŃială de asignare are aceeaşi sintaxă cu forma simplă a 
instrucŃiunii concurente de asignare, deosebirea dintre acestea rezultând din context. 

InstrucŃiunea secvenŃială de asignare a unui semnal are sintaxa următoare: 
semnal <= expresie [after întârziere]; 

 
Ca rezultat al execuŃiei acestei instrucŃiuni într-un proces, se evaluează expresia din 

dreapta simbolului de asignare şi se planifică un eveniment care constă din modificarea valorii 
semnalului. Simulatorul va modifica însă valoarea semnalului doar în momentul în care 
procesul va fi suspendat, iar în cazul în care se utilizează clauza after, după întârzierea 
specificată din acest moment. Deci, într-un proces semnalele vor fi actualizate doar după 
execuŃia tuturor instrucŃiunilor din cadrul procesului sau la întâlnirea unei instrucŃiuni wait. 

În mod tipic, sistemele de sinteză nu permit utilizarea clauzelor after, sau ignoră 
aceste clauze. Clauzele after sunt ignorate nu numai deoarece interpretarea lor pentru sinteză 
nu este specificată de standarde, ci şi pentru că ar fi dificilă garantarea rezultatelor unor 
asemenea întârzieri. De exemplu, nu este clar dacă întârzierea ar trebui interpretată ca o 
întârziere de propagare minimă sau maximă. De asemenea, nu este clar cum trebuie să 
procedeze programul de sinteză dacă o întârziere specificată în codul sursă nu poate fi 
asigurată. 

O consecinŃă a modului în care se execută asignarea semnalelor în cadrul proceselor 
este că în cazul în care există mai multe asignări a unor valori diferite la acelaşi semnal, va 
avea efect doar ultima asignare. Astfel, cele două procese din Exemplul 7 sunt echivalente. 

Exemplul 7: 
proc7: process (a) 
begin 

z <= '0'; 
z <= a; 

end process proc7; 
proc8: process (a) 

begin 
z <= a; 

end process proc8; 
 

În concluzie, trebuie să se Ńină cont de următoarele aspecte importante la utilizarea 
instrucŃiunilor de asignare a semnalelor în interiorul proceselor: 

• Orice asignare a unei valori la un semnal devine efectivă numai atunci când procesul 
este suspendat. Până în acel moment, toate semnalele îşi păstrează vechea valoare. 

• Numai ultima asignare a unei valori la un semnal va fi executată efectiv. Deci, nu are 
sens să se asigneze mai mult de o valoare la un semnal în acelaşi proces. 

 
3.1.2.2. ReacŃii inverse 

 
O altă consecinŃă a modului în care se execută asignarea semnalelor este că prin citirea 

unui semnal căruia i se asignează o valoare în acelaşi proces se va obŃine valoarea care i-a fost 
asignată semnalului la execuŃia precedentă a procesului. Citirea unui semnal şi asignarea unei 
valori la acel semnal în acelaşi proces este echivalentă cu o reacŃie inversă. Într-un proces 
combinaŃional, valoarea precedentă este o ieşire a logicii combinaŃionale, astfel încât reacŃia 
este asincronă. Într-un proces secvenŃial, valoarea precedentă este valoarea memorată într-un 
bistabil sau registru, astfel încât reacŃia este sincronă. 

În Exemplul 8 se prezintă un proces în care se utilizează reacŃia inversă sincronă. 
Procesul descrie un numărător de 4 biŃi, semnalul num fiind de tip unsigned, tip care este 
declarat în pachetul numeric_bit. Se observă că semnalul num este declarat în afara 
procesului. 

Exemplul 8: 
library ieee; 
use ieee.numeric_bit.all; 
signal num: unsigned (3 downto 0); 
process 
begin 

wait until clk = '1'; 
if reset = '1' then 

num <= "0000"; 
else 

num <= num + "0001"; 
end if; 
end process; 
 
3.1.2.3. Întârzierea inerŃială 
 
În limbajul VHDL există două tipuri de întârzieri care pot fi utilizate pentru modelarea 

sistemelor reale. Acestea sunt întârzierea inerŃială şi întârzierea de transport. MenŃionăm că 
aceste întârzieri nu se pot utiliza pentru sinteza logică. Întârzierea inerŃială este implicită, fiind 
utilizată atunci când nu se specifică tipul întârzierii.  

Clauza after presupune în mod automat întârzierea inerŃială. Într-un model cu 
întârziere inerŃială, două modificări consecutive ale valorii unui semnal de intrare nu vor 
modifica valoarea unui semnal de ieşire dacă timpul dintre aceste modificări este mai scurt 
decât întârzierea specificată. Această întârziere reprezintă inerŃia circuitului real. 

Dacă, de exemplu, apar anumite impulsuri de durată scurtă ale semnalelor de intrare, 
semnalele de ieşire vor rămâne neschimbate. 

3.1.2.4. Întârzierea de transport 
 

Întârzierea de transport trebuie indicată în mod explicit prin cuvântul cheie transport. 
Aceasta reprezintă întârzierea unei conexiuni, în care efectul apariŃiei unui impuls al unui 
semnal de intrare este propagat la ieşire cu întârzierea specificată, indiferent de durata acestui 
impuls. Întârzierea de transport este utilă în special pentru modelarea liniilor de transmisie şi a 
conexiunilor dintre componente. 

Exempu de întârzierea de transport: 
b <= transport a after 20 ns; 

 
3.1.3. Variabile 
 
RestricŃiile impuse asupra semnalelor reduc posibilităŃile de utilizare ale acestora. 

Deoarece semnalele pot păstra numai ultima valoare asignată, ele nu pot fi utilizate pentru 
memorarea rezultatelor intermediare sau temporare în cadrul unui proces. Un alt inconvenient 
este că noile valori sunt asignate semnalelor nu în momentul execuŃiei instrucŃiunii de 
asignare, ci după suspendarea execuŃiei procesului. Aceasta determină ca analiza unei 
descrieri să fie dificilă. 

Spre deosebire de semnale, variabilele pot fi declarate în interiorul unui proces şi se 
pot utiliza pentru memorarea rezultatelor intermediare. Variabilele pot fi utilizate însă numai 
în domeniul secvenŃial al limbajului VHDL, deci în interiorul proceselor sau subprogramelor, 
şi nu pot fi declarate sau utilizate direct într-o arhitectură. Ele sunt deci locale procesului sau 
subprogramului respectiv. 
 

3.1.3.1. Declararea şi iniŃializarea variabilelor 
 
Ca şi semnalele, variabilele trebuie declarate înainte de a fi utilizate. DeclaraŃia unei 

variabile este similara cu cea a unui semnal, cu deosebirea că se utilizează cuvântul cheie 
variable. În mod opŃional, în cazul variabilelor scalare se poate specifica un domeniu restrâns, 
iar în cazul variabilelor de tip tablou se poate specifica un index restrâns. Pentru ambele tipuri 
de variabile, se poate specifica o valoare iniŃială. Exemplul 9 ilustrează declararea şi 
iniŃializarea variabilelor. 

Exemplul 9: 
variable a, b, c: bit; 
variable x, y: integer; 
variable index range 1 to 10 := 1; 
variable t_ciclu: time range 10 ns to 50 ns := 10 ns; 
variable mem: bit_vector (0 to 15); 

 
Unei variabile i se atribuie o valoare iniŃială în faza de iniŃializare a simulării. Această 

valoare iniŃială este fie cea specificată în mod explicit la declararea variabilei, fie o valoare 
implicită, care este valoarea cea mai din stânga a tipului. De exemplu, pentru tipul bit valoarea 
iniŃială implicită este '0', iar pentru tipul integer această valoare este -2.147.483.647. 

Pentru sinteză, nu există o interpretare hardware a valorilor iniŃiale, astfel încât 
sistemele de sinteză ignoră valorile iniŃiale sau semnalează eroare. 

3.1.3.2. InstrucŃiunea de asignare a variabilelor 
 

O instrucŃiune de asignare a unei variabile înlocuieşte valoarea curentă a variabilei cu 
o nouă valoare specificată de o expresie. Expresia poate conŃine variabile, semnale şi literale. 
Variabila şi rezultatul evaluării expresiei trebuie să fie de acelaşi tip. InstrucŃiunea de asignare 
a unei variabile are sintaxa următoare: 

variabilă := expresie; 
Această instrucŃiune este similară cu asignările din majoritatea limbajelor de 

programare. Spre deosebire de o instrucŃiune secvenŃială de asignare a semnalelor, asignarea 
unei variabile este executată instantaneu, deci într-un timp de simulare zero. 

Există următoarele diferenŃe principale între asignarea semnalelor şi a variabilelor: 
• În cazul asignării semnalelor, se planifică un eveniment pentru actualizarea valorii 

acestora, iar actualizarea se execută numai la suspendarea procesului, în timp ce pentru 
asignarea variabilelor nu se planifică un eveniment, iar actualizarea se execută 
instantaneu. 

• La asignarea semnalelor se poate specifica o întârziere, în timp ce asignarea 
variabilelor nu poate fi întârziată. 

• Într-un proces, doar ultima asignare a unei valori la un semnal este efectivă. În schimb, 
pot exista numeroase asignări la o variabilă în acelaşi proces, toate fiind efective. 
 
Exemplul 10 ilustrează utilizarea variabilelor pentru memorarea rezultatelor 

intermediare. 
Exemplul 10 
entity add_1 is 

port (a, b, cin: in bit; 
s, cout: out bit); 

end add_1; 
architecture functional of add_1 is 
begin 
process (a, b, cin) 

variable s1, s2, c1, c2: bit; 
begin 

s1 := a xor b; 
c1 := a and b; 
s2 := s1 xor cin; 
c2 := s1 and cin; 
s <= s2; 
cout <= c1 or c2; 

end process; 
end functional; 

 
Exemplul anterior descrie un sumator elementar. Acesta nu este modul optim de a 

descrie un sumator elementar, dar ilustrează utilizarea variabilelor. Rezultatul se generează 
prin crearea a două semisumatoare, primul generând ieşirile s1 şi c1, iar al doilea generând 
ieşirile s2 şi c2. În final, ieşirile sunt asignate celor două porturi de ieşire s şi cout prin 
instrucŃiuni de asignare a semnalelor, deoarece porturile sunt semnale. 

3.1.4. InstrucŃiunea if 
3.1.4.1. Sintaxa şi execuŃia instrucŃiunii if 

 
InstrucŃiunea if selectează pentru execuŃie una sau mai multe secvenŃe de instrucŃiuni, 

în funcŃie de valoarea unei condiŃii corespunzătoare secvenŃei respective. Sintaxa acestei 
instrucŃiuni este următoarea: 

if condiŃie then secvenŃă_de_instrucŃiuni 

[elsif condiŃie then secvenŃă_de_instrucŃiuni ...] 
[else secvenŃă_de_instrucŃiuni] 
end if; 

 
Fiecare condiŃie este o expresie booleană, care se evaluează la valoarea TRUE sau 

FALSE. Pot exista mai multe clauze elsif, dar poate exista o singură clauză else. Se evaluează 
mai întâi condiŃia de după cuvântul cheie if, şi dacă este adevărată, se execută secvenŃa de 
instrucŃiuni corespunzătoare. În caz contrar, dacă este prezentă clauza elsif, se evaluează 
condiŃia de după această clauză şi dacă această condiŃie este adevărată, se execută secvenŃa de 
instrucŃiuni corespunzătoare. În caz contrar, dacă sunt prezente alte clauze elsif, se continuă 
cu evaluarea condiŃiei acestor clauze. Dacă nici o condiŃie evaluată nu este adevărată, se 
execută secvenŃa de instrucŃiuni corespunzătoare clauzei else, dacă aceasta este prezentă. 

Exemplul 11: 
process (a, b) 
begin 

if a = b then 
rez <= 0; 

elsif a < b then 
rez <= -1; 

else 
rez <= 1; 

end if; 
end process; 

 
3.1.4.2. Interpretarea sintezei instrucŃiunii if 

 
Exemplul 12 prezintă utilizarea unei asemenea instrucŃiuni pentru descrierea unui 

comparator. 
Exemplul 12: 
library ieee; 
use ieee.numeric_bit.all; 
 
entity comp is 

port (a, b: in unsigned (7 downto 0); 
egal: out bit); 

end comp; 
architecture functional of comp is 
begin 
process (a, b) 
begin 

if a = b then 
egal <= '1'; 

else 
egal <= '0'; 

end if; 
end process; 
end functional; 

 
În exemplul anterior, se testează egalitatea a două semnale de tip unsigned, 

reprezentând două valori întregi de câte 8 biŃi, rezultând o valoare de tip bit. 
O instrucŃiune if cu ramuri multiple, în care apare cel puŃin o clauză elsif, este 

implementată prin mai multe etaje de multiplexoare. Considerăm instrucŃiunea if din 
Exemplul 13. 

Exemplul 13: 
process (a, b, c, s0, s1) 
begin 

if s0 = '1' then 
z <= a; 

elsif s1 = '1' then 
z <= b; 

else 
z <= c; 

end if; 
end process; 

 
CondiŃiile din ramurile succesive ale unei instrucŃiuni if sunt evaluate independent. În 

exemplul anterior, condiŃiile implică cele două semnale s0 şi s1. Pot exista oricâte condiŃii, 
fiecare din acestea fiind independent de celelalte. Structura instrucŃiunii if asigură faptul că 
primele condiŃii vor fi testate înaintea celorlalte. În acest exemplu, semnalul s0 a fost testat 
înaintea semnalului s1. Această prioritate se reflectă în circuitul rezultat, în care multiplexorul 
controlat de semnalul s0 este mai apropiat de ieşire decât multiplexorul controlat de semnalul 
s1. 

Este important să se reŃină existenŃa acestei priorităŃi cu care se testează condiŃiile, 
astfel încât să fie eliminate condiŃiile redundante. Considerăm instrucŃiunea if din Exemplul 
14, care este echivalentă cu instrucŃiunea if din Exemplul 13. 

Exemplul 14: 
process (a, b, c, s0, s1) 
begin 

if s0 = '1' then 
z <= a; 

elsif s0 = '0' and s1 = '1' then 
z <= b; 

else 
z <= c; 

end if; 
end process; 

 
CondiŃia suplimentară s0 = '0' este redundantă, deoarece ea va fi testată numai dacă 

prima condiŃie a  instrucŃiunii if este falsă. Se recomandă evitarea unor asemenea redundanŃe, 
deoarece nu există garanŃia că ele vor fi detectate şi eliminate de către sistemul de sinteză. 

În cazul instrucŃiunilor if cu ramuri multiple, în mod normal fiecare condiŃie va fi 
dependentă de semnale şi variabile diferite. Dacă fiecare ramură este dependentă de acelaşi 
semnal, atunci este mai avantajos să se utilizeze o instrucŃiune case.  

 
3.1.4.3. InstrucŃiuni if incomplete 

 
În exemplele prezentate până acum, toate instrucŃiunile if au fost complete. Cu alte 

cuvinte, semnalului destinaŃie i s-a asignat o valoare în toate condiŃiile posibile. Sunt posibile 
însă două situaŃii în care unui semnal nu i se asignează o valoare: atunci când lipseşte clauza 
else a instrucŃiunii if sau atunci când unui semnal nu i se asignează o valoare într-o ramură a 
instrucŃiunii if. În ambele cazuri, interpretarea este aceeaşi. În situaŃiile în care unui semnal nu 
i se asignează o valoare, semnalul îşi păstrează valoarea precedentă. 

Se pune însă problema care este valoarea precedentă a semnalului. Dacă există o 
instrucŃiune anterioară de asignare în care semnalul apare ca destinaŃie, valoarea precedentă 
provine de la acea instrucŃiune de asignare. În caz contrar, valoarea provine din execuŃia 
precedentă a procesului, ceea ce conduce la existenŃa unei reacŃii inverse în cadrul circuitului. 
Primul caz este ilustrat prin Exemplul 15. 

Exemplul 15: 
process (a, b, enable) 
begin 

z <= a; 
if enable = '1' then 

z <= b; 
end if; 

end process; 
 

În acest caz, instrucŃiunea if este incompletă deoarece lipseşte clauza else. În 
instrucŃiunea if, semnalul z primeşte o valoare în cazul în care condiŃia enable = '1' este 
adevărată, dar rămâne la valoarea precedenta în cazul în care condiŃia este falsă. Valoarea 
precedentă provine din instrucŃiunea de asignare aflată înaintea instrucŃiunii if. 
InstrucŃiunea if din Exemplul 15 este echivalentă cu instrucŃiunea if din exemplul următor. 

Exemplul 16: 
process (a, b, enable) 
begin 

if enable = '1' then 
z <= b; 

else 
z <= a; 

end if; 
end process; 

 
În cazul în care instrucŃiunea if este incompletă şi nu există o instrucŃiune anterioară de 

asignare, va exista o reacŃie inversă de la ieşirea circuitului la intrare. Aceasta deoarece 
valoarea semnalului de la execuŃia precedentă a procesului este păstrată şi aceasta devine 
valoarea semnalului la execuŃia curentă a procesului. 

O asemenea formă a instrucŃiunii if este utilizată pentru a descrie un bistabil sau un 
registru cu o intrare de validare, ca în Exemplul 17. 

Exemplul 17: 
process 
begin 

wait until clk = '1'; 
if en = '1' then 

q <= d; 
end if; 

end process; 
 

Semnalul q este actualizat cu noua valoare a semnalului d în cazul în care condiŃia este 
adevărată, dar nu este actualizat în cazul în care condiŃia este falsă. În acest caz, valoarea 
precedentă a semnalului q este păstrată prin reacŃia inversă secvenŃială a semnalului q.  

InstrucŃiunea if din Exemplul 17 este echivalentă cu instrucŃiunea if completă din 
exemplul următor. 

Exemplul 18: 
process 
begin 

wait until clk = '1'; 
if en = '1' then 

q <= d; 
else 

q <= q; 
end if; 

end process; 
 

În cazul în care condiŃia este falsă, se asignează semnalului q valoarea anterioară a 
acestuia, ceea ce este echivalent cu păstrarea valorii sale anterioare. 

Una din cele mai obişnuite erori întâlnite în descrierile VHDL destinate sintezei logice 
este introducerea neintenŃionată a reacŃiei inverse în circuit datorită unei instrucŃiuni if 
incomplete. Pentru a evita introducerea reacŃiei inverse, proiectantul trebuie să se asigure că 
fiecare semnal căruia i se asignează o valoare într-un process (care este deci un semnal de 
ieşire din proces) primeşte o valoare în fiecare combinaŃie posibilă a condiŃiilor.  

În practică, există două posibilităŃi pentru aceasta: asignarea unei valori unui semnal 
de ieşire în fiecare ramură a instrucŃiunii if şi includerea clauzei else, sau iniŃializarea 
semnalului printr-o instrucŃiune de asignare necondiŃionată înaintea instrucŃiunii if. 

 
3.1.4.4. InstrucŃiuni if în care apar variabile 

 
Până acum, în instrucŃiunile if au fost utilizate doar semnale. Aceleaşi reguli se aplică 

şi în cazul utilizării variabilelor, cu o singură diferenŃă. Ca şi în cazul unui semnal, dacă unei 
variabile i se asignează valori numai în anumite ramuri ale instrucŃiunii if, se păstrează 
valoarea anterioară a variabilei prin reacŃie inversă. Spre deosebire de cazul utilizării unui 
semnal, citirea şi scrierea unei variabile în acelaşi proces va determina o reacŃie inversă numai 
dacă citirea apare înaintea scrierii. În acest caz, valoarea citită este valoarea precedentă a 
variabilei. În cazul citirii şi scrierii în acelaşi proces a unui semnal, va rezulta întotdeauna o 
reacŃie inversă. 

Această observaŃie se poate utiliza pentru a crea registre sau numărătoare utilizând 
variabile. Reamintim că un proces secvenŃial este interpretat la sinteză prin plasarea unui 
bistabil sau registru înaintea ieşirii fiecărui semnal căruia i s-a asignat o valoare în procesul 

respectiv. Aceasta înseamnă că în mod normal variabilele nu sunt înscrise în bistabile sau 
registre. Dacă există însă o reacŃie inversă a valorii precedente a unei variabile, această reacŃie 
se realizează printr-un bistabil sau registru pentru ca procesul să fie sincron. 

În Exemplul 21 se descrie un numărător utilizând tipul întreg unsigned. La 
incrementarea unei valori de tip unsigned, dacă valoarea este cea maximă a domeniului se 
obŃine valoarea minimă a domeniului. 

Exemplul 21: 
process 
variable num: unsigned (7 downto 0); 
begin 

wait until clk = '1'; 
if reset = '1' then 

num := "00000000"; 
else 

num := num + 1; 
end if; 
rez <= num; 

end process; 
 

În acest exemplu, în ramura else a instrucŃiunii if se citeşte valoarea precedentă a 
variabilei num pentru a calcula valoarea următoare. Rezultă astfel o reacŃie inversă. 

De notat că în acest exemplu se crează de fapt două registre. Conform regulilor pentru 
reacŃia inversă, variabila num va fi înscrisă într-un registru. Semnalul rez va fi de asemenea 
înscris într-un registru, deoarece toate semnalele cărora li se asignează o valoare într-un 
proces secvenŃial vor fi înscrise în registre. Acest registru suplimentar va conŃine întotdeauna 
aceeaşi valoare ca şi registrul variabilei num. Sistemul de sinteză va elimina în mod normal 
acest registru redundant. 

 
3.2 Teme propuse 

 
3.2.1 AnalizaŃi exemplele prezentate care se referă la instrucŃiunile secvenŃiale. 

SimulaŃi descrierea comparatorului (Exemplul 12), a unui bistabil (Exemplul 18), a 
numărătorului (Exemplul 21). UtilizaŃi sistemul Active-HDL pentru compilarea cu succes a 
acestor descrieri. 

3.2.2 DescrieŃi un multiplexor 4:1 de 4 biŃi utilizând o instrucŃiune IF. Intrările 
multiplexorului sunt a[3:0], b[3:0], c[3:0], d[3:0], s[1:0], iar ieşirile sunt q[3:0]. 

 
 


